Wie können Textmarkennamen direkt im Dokument angezeigt oder ausgedruckt werden?

· Zurück zum Hauptindex
Die Namen der Textmarken im Dokument erkenne ich nur, wenn ich den Dialog Einfügen > Textmarke… aufrufe. Ich möchte die Textmarkennamen aber direkt im Dokument angezeigt und wenn möglich auch ausgedruckt haben.

Diese Seite

· Einleitung
· Textmarkennamen direkt im Dokument anzeigen
· Textmarkennamen als Kommentare («Markups») in Sprechblasen anzeigen
· Textmarkennamen als Verzeichnis anzeigen
· Links zu diesem Thema
Einleitung

· Zurück zum Anfang
Der im Vorspann geäusserte Wunsch kann ausschliesslich mit Hilfe eines Makros befriedigt werden. In diesem Skript finden Sie drei praxistaugliche Vorschläge dazu. Jede dieser Lösungen in Form einer VBA-Routinen führt zu einer anderen Darstellungsweise der Textmarkennamen.

Textmarkennamen direkt im Dokument anzeigen

Bei dor Verwendung dieser Routine werden
de Hamen der Towmarken dickdt im
Dokument mit spezieller Formationung dar-
gestellt.



Zurück zum Anfang
Das erste Makro mit dem Namen TextmarkenImTextAnzeigen schreibt die Textmarkennamen jeweils an jene Stelle im Dokument, an der die Textmarken erscheinen und zwar als einfachen Text. Dabei heben sich die Namen durch eine spezielle Formatierung vom eigentlichen Text ab. Diese Formatierung können Sie über die Formatvorlage mit der Bezeichnung textmarke jederzeit anpassen. Selbstverständlich werden diese Textmarkennamen auf Wunsch auch ausgedruckt.

Der Code beinhaltet auch ein kleines Makro mit dem Namen TextmarkenImTextLöschen, welches die eingefügten Textmarkennamen auf Wunsch wieder zum verschwinden bringt.

VBA-Code

· So bringen Sie das Makro ein und verknüpfen es mit einem neuen Schaltknopf auf der Symbolleiste Ihrer Wahl


Sub TextmarkenImTextAnzeigen()

  Dim oDoc As Document, oBM As Bookmark, oStyle As Style, oRange As Range

  Dim rc As Long, tmp As String
  Set oDoc = ActiveDocument

  On Error Resume Next
  Set oStyle = ActiveDocument.Styles("textmarke")

  rc = Err.Number

  On Error GoTo 0

  If Not rc = 0 Then
    Set oStyle = oDoc.Styles.Add(Name:="textmarke", Type:=wdStyleTypeCharacter)

    oStyle.Font.Name = "Verdana"

    oStyle.Font.Size = 8.5

    oStyle.Font.ColorIndex = wdTeal

  End If
  TextmarkenImTextLöschen

  Application.ScreenUpdating = False
  For Each oBM In oDoc.Bookmarks

    Set oRange = oBM.Range

    If oRange.Characters.Last.Next = " " Then
      tmp = " " & oBM.Name

    Else
      tmp = oBM.Name & " "

    End If
    oRange.InsertAfter tmp

    oRange.SetRange oRange.End - Len(oBM.Name) - 1, oRange.End

    oRange.Style = "textmarke"

  Next
  Application.ScreenUpdating = True
End Sub
Sub TextmarkenImTextLöschen()

  Application.ScreenUpdating = False
  With ActiveDocument.Range.Find

    .Style = "textmarke"

    .Replacement.Text = ""

    .Execute Replace:=wdReplaceAll

  End With
  Application.ScreenUpdating = True
End Sub
(Die Textmarkennamen werden direkt als Text im Dokument eingefügt und mit der Formatvorlage, welche die Bezeichnung «textmarke» trägt, formatiert.

